
HP-35S Calculator Program     Utilities 2A 

Azimuth and Distance from Co-ordinates (‘Inverse’) 

Programmer: Dr. Bill Hazelton 

Date: September, 2008. Version:   1.0  Mnemonic: I for ‘Inverse’ Program 
 
Line Instruction Display User Instructions 
I001 LBL I   LBL I 
I002 CLSTK   CLEAR  5 
I003 SF  10    FLAGS  1  .0 
I004 COORD  INVERSE  (Key in as EQN RCL C, RCL O,  etc.; ENTER to end) 
I005 PSE    PSE 
I006 NTR  FAR  POINT  (Key in as EQN RCL N, RCL T,  etc.; ENTER to end) 
I007 PSE    PSE 
I008 XEQ  I029   
I009 RCL  P   
I010 STO  F    STO  F 
I011 NTR  NEAR  POINT  (Key in as EQN RCL N, RCL T,  etc.; ENTER to end) 
I012 PSE    PSE 
I013 XEQ  I029   
I014 RCL  P   
I015 STO  N    STO  N 
I016 RCL  F   
I017 RCL–  N   
I018 STO  V    STO  V 
I019 ARG    ARG 
I020 x ≥ 0?    x ? 0   5 
I021 GTO  I024   
I022 360   
I023 +   
I024 →HMS    →HMS 
I025 RCL  V   
I026 ABS    ABS 
I027 STOP  Press R/S 
I028 GTO  I006   
I029 CLx    CLEAR  1 
I030 STO  N    STO  N 
I031 STO  E    STO  E 
I032 KEY  IN  N  (Key in as EQN RCL K, RCL E,  etc.; ENTER to end) 
I033 PSE    PSE 
I034 INPUT  N    INPUT  N 
I035 KEY  IN  E  (Key in as EQN RCL K, RCL E,  etc.; ENTER to end) 
I036 PSE    PSE 
I037 INPUT  E    INPUT  E 
I038 RCL  E   
I039 0  i 1  (Key in as 0, then i, then 1, press ENTER.) 

I040 ×   
I041 RCL+  N   
I042 STO  P    STO  P 
I043 RTN    RTN 
 
This program is designed to hide completely the complex number work that the calculator 
performs to compute the azimuth and distance from the co-ordinates. All values entered and 
returned are presented in much the same manner as with the equivalent operation (Rectangular to 
Polar) in the HP-33S calculator. 

Notes 

 (1) Set the calculator into DEGREES mode (press MODE 1) before starting, to make 
sure that you get degrees, minutes and seconds for the azimuth. 

 (2) This is a basic co-ordinate ‘inverse’ program that computes the azimuth and 
distance between two points whose co-ordinates are supplied. While the program 
works with complex number representations internally, this is hidden from the 
user. 

 (3) Azimuths are by themselves in HP notation, i.e., DDD.MMSSss. 

 (4) In order to display the prompts, this program sets Flag 10. However, the program 
never formally ends, because it is up to the user to decide when to stop and move 
control elsewhere. So the program never clears Flag 10. If you require Flag 10 to 
be clear, in order to process equations, you must do this manually. 

Theory 

If the co-ordinates of two points are given, there are two ways by which the azimuth and distance 
between them may be derived. These are equivalent, and will produce the same result. This HP-
35s program uses the second method, viz., vectors stored as complex numbers. 

 

1. Using co-ordinates alone, as in the figure above the differences between the easting co-
ordinates of the points (∆E) and difference between the northing co-ordinates of the two points 
(∆N) are obtained. The point from which the azimuth is desired to come is termed the ‘near’ 
point, while the point to which the azimuth points is termed the ‘far’ point. So: 

 ∆E  =  Eƒ  –  En 

 ∆N  =  Nƒ  –  Nn 

The distance, d, between the two points is computed using: 

 d  =  ΔE2 + ΔN2  

The azimuth, θ, between the two points is computed using: 

 θ  =  arctan
∆ E
∆ N

⎛
⎝⎜

⎞
⎠⎟  

The ATAN2 function (in some programming languages and Excel) takes the two components, ∆E 
and ∆N, as separate arguments, and returns an azimuth in the range –180° to +180°. This is easily 
converted to an azimuth in the range 0° to 360° The HP calculator function R→P performed a 
similar process, but it is not present on the HP-35s, so the angle must be put into its correct 
quadrant manually, as the ATAN function returns a value between –90° and +90°. 

 

2. Using vectors, as shown in the figure above, the co-ordinates of each point are entered as 
the components of a pair of 2-D vectors. In the HP-35s, 2-D vectors are best handled as complex 
numbers, so that the co-ordinates are stored as the two vectors from the origin, vf and vn, as 
follows: 

 vf  =  Nƒ  +  i  Eƒ 

 vn  =  Nn  +  i  En 

The vector from the near point to the far point, vx, is then the difference between the two vectors: 

 vx  =  vf  –  vn 

This calculation is handled by a simple vector (albeit using complex numbers) subtraction in the 
HP-35s. 

To obtain the length of the vector vx, its absolute value must be taken (using the ABS function): 

 d  =  v x  

The azimuth of the vector vx is calculated as the argument of the vector, using the ARG function: 

 θ  =  arg(vx) 

The HP-35s returns the azimuth in the units in which the calculator us currently set, usually 
degrees. The value will be in the range –180° to +180°. The program brings this value into the 
range 0° to 360°, and converts it to degrees, minutes and seconds, in HP notation (i.e., 
HHH.MMSSsss). 

Sample Computations and Running the Program 

Example 1 

 

In the above example, ∆E  =  350 and ∆N  =  150. Using the first method discussed above: 

 d  =  ΔE2 + ΔN2   =  122,500 + 22,500   =  145,000   =  380.789  

 θ  =  arctan
∆ E
∆ N

⎛
⎝⎜

⎞
⎠⎟   =  arctan

350
150

⎛
⎝⎜

⎞
⎠⎟   =  arctan 2.333…( )   =  66° 48' 05"  

Using the second method, the complex numbers formed are: 

 vf  =  2,850  +  i  850 

 vn  =  2,700  +  i  500 

and the difference between them is: 

 vx  =  150  +  i  350 

so d  =  v x   =  3502 + 1502   =  122,500 + 22,500   =  145,000   =  380.789  

and θ  =  arg(vx)  =  66° 48' 05" 

Using the calculator for this process, begin by pressing XEQ  I  and then the Enter key. 

The calculator briefly displays COORD  INVERSE, to indicate the program, then NTR  FAR  
POINT, to tell the user to enter the co-ordinates of the far point, and finally KEY  IN  N. The 
calculator then prompts for entry of the northing co-ordinate of the far point, as follows: 

 N? 

 0.0000 

Key in the northing of the far point, in this case 2850, and press R/S. 

The calculator briefly displays KEY  IN  E, then prompts for keying in the easting of the far 
point: 

 E? 
 0.0000 

Key in the easting of the far point, in this case 850, and press R/S. 

The calculator briefly displays NTR  NEAR  POINT, then KEY  IN  N, then prompts the user to 
key in the northing of the near point, displaying: 

 N? 
 0.0000 

Key in the northing of the near point, in this case 2700, and press R/S. 

The calculator briefly displays KEY  IN  E, then prompts for keying in the easting of the near 
point: 

 E? 
 0.0000 

Key in the easting of the near point, in this case 500, and press R/S. 

The calculator then displays the azimuth of the line in line 1 (the upper line of the display) in HP 
notation (DDD.MMSSsss), and the length of the line in line 2 (the lower line of the display. In 
this case, the display shows: 

 66.4805    (meaning 66° 48' 05", the azimuth of the line) 
 380.7887   (the length of the line) 

At this point, if the user presses R/S, the program prompts for the northing of the far point, as 
discussed above. Otherwise, the user can keep working on whatever they were doing before. Note 
that Flag 10 will now be set. 

Example 2 

Near Point E     500   Far Point E       850 
  N  2,700     N    2,600 

Begin by pressing XEQ  I  and then the Enter key. 

The calculator briefly displays COORD  INVERSE, to indicate the program, then NTR  FAR  
POINT, to tell the user to enter the co-ordinates of the far point, and finally KEY  IN  N. The 
calculator then prompts for entry of the northing co-ordinate of the far point, as follows: 

 N? 
 0.0000 

Key in the northing of the far point, in this case 2600, and press R/S. 

The calculator briefly displays KEY  IN  E, then prompts for keying in the easting of the far 
point: 

 E? 
 0.0000 

Key in the easting of the far point, in this case 850, and press R/S. 

The calculator briefly displays NTR  NEAR  POINT, then KEY  IN  N, then prompts the user to 
key in the northing of the near point, displaying: 

 N? 
 0.0000 

Key in the northing of the near point, in this case 2700, and press R/S. 

The calculator briefly displays KEY  IN  E, then prompts for keying in the easting of the near 
point: 

 E? 
 0.0000 

Key in the easting of the near point, in this case 500, and press R/S. 

The calculator then displays the azimuth of the line in line 1 (the upper line of the display) in HP 
notation (DDD.MMSSsss), and the length of the line in line 2 (the lower line of the display. In 
this case, the display shows: 

 105.5643   (meaning 105° 56' 43", the azimuth of the line) 
 364.0055   (the length of the line) 

At this point, if the user presses R/S, the program prompts for the northing of the far point, as 
discussed above. Otherwise, the user can keep working on whatever they were doing before. Note 
that Flag 10 will now be set. 

Distance  =  364.0055    Azimuth  =  105° 56' 43" 

Example 3 

Near Point E     500   Far Point E       450 
  N  2,700     N    2,500 

Process the co-ordinates through the calculator as above, to get the following results: 

Distance  =  206.1553    Azimuth  =  194° 02' 10" 

Example 4 

Near Point E     500   Far Point E       300 
  N  2,700     N    2,800 

Process the co-ordinates through the calculator as above, to get the following results: 

Distance  =  223.6068    Azimuth  =  296° 33' 54" 

 

Storage Registers Used 

E Easting co-ordinates of the entered points. 

F Co-ordinates of far point, as a complex number. 

N Northing co-ordinate of the entered points, then the co-ordinates of the near point, as a 
complex number. 

P Temporary storage of co-ordinates of entered point, as a complex number. 

V Vector from the near point to the far point, stored as a complex number. 

Statistical Registers: not used. 

Labels Used 

Label I  Length  =   191  Checksum  =  BA0E 

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. 
Use the sample computation to check proper operation after entry. 

Routines Called 

None. The program does call the same segment of code twice, for co-ordinate entry, but this is 
within the program. 

Flags Used 

The program sets Flag 10, to allow the display of prompts. However, as the program never 
formally ends (it is used by the user as needed), it never clears this flag. If the flag is needed to be 
cleared, clear it manually. 

 


